Chapter 11: Mouse Control – Moving Things Around.

This chapter will show you how to make your program respond to a mouse. There are two different ways to use the mouse: tracking mode and clicking mode. Both are discussed with sample programs.

Tracking Mode:

In mouse tracking mode, there are three numeric functions (**mousex**, **mousey**, and **mouseb**) that will return the coordinates of the mouse pointer over the graphics output area. If the mouse is not over the graphics display area then the mouse movements will not be recorded (the last location will be returned).


```
1     # mousetrack.kbs
2     # track the mouse with a circle
3
4     print "Move the mouse around the graphics window."
5     print "Click left mouse button to quit."
6
7     fastgraphics
8
9     # do it over and over until the user clicks left
10    while mouseb <> MOUSEBUTTON_LEFT
11         # erase screen
12         clg
13         # draw new ball
14         color red
15         circle mousex, mousey, 10
16         refresh
17    end while
18
```

```
19    print "all done."
20    end
```

Program 70: Mouse Tracking



Sample Output 70: Mouse Tracking

New Concept

```
mousex or mousex()
mousey or mousey()
mouseb or mouseb()
```

The three mouse functions will return the current location of the mouse as it is moved over the graphics display area. Any mouse motions outside the graphics display area are not recorded, but the last known coordinates will be returned.

mousex	Returns the x coordinate of the mouse pointer position. Ranges from 0 to **graphwidth** -1.
mousey	Returns the y coordinate of the mouse pointer position. Ranges from 0 to **graphheight** -1.
mouseb	<table><tr><td>0 or MOUSEBUTTON_NONE</td><td>Returns this value when no mouse button is being pressed.</td></tr><tr><td>1 or MOUSEBUTTON_LEFT</td><td>Returns this value when the "left" mouse button is being pressed.</td></tr><tr><td>2 or MOUSEBUTTON_RIGHT</td><td>Returns this value when the "right" mouse button is being pressed.</td></tr><tr><td>4 or MOUSEBUTTON_CENTER</td><td>Returns this value when the "center" mouse button is being pressed.</td></tr></table> If multiple mouse buttons are being pressed at the same time then the value returned will be the button values added together.

Clicking Mode:

The second mode for mouse control is called "Clicking Mode". In clicking mode, the mouse location and the button (or combination of buttons) are stored when the click happens. Once a click is processed by the program a *clickclear* command can be executed to reset the click, so the next one can be recorded.

```
1     # mouseclick.kbs
2     # X marks the spot where you click
3
4     print "Move the mouse around the graphics window"
5     print "click left mouse button to mark your spot"
6     print "click right mouse button to stop."
7     clg
8     clickclear
9     while clickb <> MOUSEBUTTON_RIGHT
10         # clear out last click and
11         # wait for the user to click a button
12         clickclear
13         while clickb = MOUSEBUTTON_NONE
14              pause .01
15         end while
16         #
17         color blue
18         stamp clickx, clicky, 5, {-1,-2, 0,-1, 1,-2, 2,-
      1, 1,0, 2,1, 1,2, 0,1, -1,2, -2,1, -1,0, -2,-1}
19    end while
20    print "all done."
21    end
```

Program 71: Mouse Clicking



Sample Output 71: Mouse Clicking


```
clickx  or clickx()
clicky  or clicky()
clickb or clickb()
```

The values of the three click functions are updated each time a mouse button is clicked when the pointer is on the graphics output area. The last location of the mouse when the last click was received are available from these three functions.

clickclear

The **clickclear** statement resets the **clickx**, **clicky**, and **clickb** functions to zero so that a new click will register when **clickb** <> 0.

The big program this chapter uses the mouse to move color sliders so that we can see all 16,777,216 different colors on the screen.


```
1     # colorchooser.kbs
2     fastgraphics
3
4     print "colorchooser - find a color"
5     print "click and drag red, green and blue sliders"
6
7     # variables to store the color parts
8     r = 128
9     g = 128
10    b = 128
11
12    call display(r,g,b)
13
14    while true
15         # wait for click
16         while mouseb = 0
17              pause .01
```

```
18        end while
19        # change color sliders
20        # the red slider y range is 0 >= red < 75
21        if mousey < 75 then
22             r = mousex
23             if r > 255 then r = 255
24        end if
25        # the green slider y range is 75 >= red < 150
26        if mousey >= 75 and mousey < 150 then
27             g = mousex
28             if g > 255 then g = 255
29        end if
30        # the blue slider y range is 150 >= red < 225
31        if mousey >= 150 and mousey < 225 then
32             b = mousex
33             if b > 255 then b = 255
34        end if
35        call display(r,g,b)
36   end while
37   end
38
39   subroutine colorline(r,g,b,x,y)
40        # draw part of the color bar the color r,g,b
     from x,y to x,y+37
41        color rgb(r, g, b)
42        line x, y, x, y+37
43   end subroutine
44
45   subroutine redsliderbar(r,g,b)
46        # draw the red bar from 0,0 to 255,74
47        font "Tahoma", 30, 100
48        color rgb(255, 0, 0)
49        text 260, 0, "r"
50        for t = 0 to 255
51             # red and red hues
52             call colorline(t, 0, 0, t, 0)
53             call colorline(t, g, b, t, 38)
54        next t
55        color black
```

```
56        rect r-1, 0, 3, 75
57    end subroutine
58
59    subroutine greensliderbar(r,g,b)
60        # draw thegreen bar from 0,75 to 255,149
61        font "Tahoma", 30, 100
62        color rgb(0, 255, 0)
63        text 260, 75, "g"
64        for t = 0 to 255
65            # green and green hues
66            call colorline(0, t, 0, t, 75)
67            call colorline(r, t, b, t, 113)
68        next t
69        # slider
70        color black
71        rect g-1, 75, 3, 75
72    end subroutine
73
74    subroutine bluesliderbar(r,g,b)
75        # draw the blue bar from 0,150 to 255,224
76        font "Tahoma", 30, 100
77        color rgb(0, 0, 255)
78        text 260, 150, "b"
79        for t = 0 to 255
80            # blue and blue hues
81            call colorline(0, 0, t, t, 150)
82            call colorline(r, g, t, t, 188)
83        next t
84        # slider
85        color black
86        rect b-1, 150, 3, 75
87    end subroutine
88
89    subroutine display(r, g, b)
90        clg
91        call redsliderbar(r,g,b)
92        call greensliderbar(r,g,b)
93        call bluesliderbar(r,g,b)
94        # draw swatch
```

```
95          color rgb(r,g,b)
96          rect 151,226,150,75
97          refresh
98          # draw the RGB values
99               color black
100         font "Tahoma", 13, 100
101         text 5, 235, "(" + r + "," + g + "," + b + ")"
102   end subroutine
```

Program 72: Big Program - Color Chooser



Sample Output 72: Big Program - Color Chooser

Exercises:

r f m t x v t x n j
j a a o h k s f o u
n c e y u t c l e c
b e x l e s h i y l
k n z m c s e w l i
c t m o r k u b k c
i e z u n i c o g k
l r p s g s g i m y
c j i e h w l h l m
c x l x m f z a t c

center, clickb, clickclear, clickx, clicky, left, mouseb, mousex, mousey, right

1. Create a program that will draw a series of connected lines and display the points on the screen as the lines are drawn.

When the left button of the mouse is clicked draw a small circle, print the coordinates, draw a line to the previous coordinates (if not the first point), and remember the point so that it can be the start of the next line. Repeat this until the user clicks stop.

```
46,62
187,59
178,132
108,96
```



2. Create a program that will allow the user to use the mouse like a paintbrush. When the user has the left button depressed then plot a point at that location. To make the line wider you may draw a circle with a radius of 2 or 3.

For extra skill when the user presses the right button make the pen color a random color

3. Use the smiling face subroutine from Problem 1 in the subroutines chapter to make a mouse drawing program with the smile. When the user clicks on a point of the screen draw a face there.